Microprocessors & Interfacing

AVR Programming (IV)

Lecturer : Annie Guo

Lecture Overview

- Stack and stack operation
- Assembly function and function call
  - Calling convention
  - Examples

Stack

- What is stack?
  - A data structure in which a data item that is Last In is First Out (LIFO)
- In AVR, a stack is implemented as a block of consecutive locations in the data memory
- A stack has at least two parameters:
  - Bottom
  - Stack pointer (SP)



Stack Bottom

- The stack usually *grows from high addresses to low addresses*
- The stack bottom is the location with the highest address in the stack
- In AVR, 0x0200 is the lowest address for stack
  - i.e. stack bottom >=0x0200



Stack Pointer

- In AVR, the stack pointer, *SP*, is an I/O register pair, *SPH:SPL*, they are defined in the device definition file
  - m2560def.inc
- Default value of the stack pointer is 0x21FF
- The stack pointer always points to the top of the stack
  - Definition of the stack top varies:
    - the location of the Last-In element;
      - E.g, in 68K
    - the location available for the next element to be stored
      - E.g. in AVR

Stack Operations

- There are two stack operations:
  - Push
    - Implemented by instruction *PUSH*
  - Pop
    - Implemented by instruction *POP*

PUSH

- Syntax: ***push Rr***
- Operands: Rr$\in${r0, r1, …, r31}
- Operation: (SP) $\leftarrow$ Rr
  SP $\leftarrow$ SP – 1
- Words: 1
- Cycles: 2

POP

- Syntax: ***pop Rd***
- Operands: Rd$\in${r0, r1, …, r31}
- Operation: SP $\leftarrow$ SP + 1
  Rd $\leftarrow$ (SP)
- Words: 1
- Cycles: 2

Functions

- Stack is used in function calls
- Functions are used
  - in top-down design
    - Conceptual decomposition - easy to design
  - for modularity
    - Readability and maintainability
  - for reuse
    - Design once and use many times
      - Common code with parameters
    - **Store once** and use many times
      - Saving code size, hence memory space

C Code Example

```
unsigned int pow(unsigned int b, unsigned int e) {          // int parameters b & e,
                                                            // returns an integer
        unsigned int i, p;                                  // local variables
        p = 1;
        for (i=0; i<e; i++)                                 // p = b^e
                p = p*b;
        return p;                                   // return value of the function
}

int main(void) {
        unsigned int m, n;
        m = 2;
        n = 3;
        m = pow(m, n);
        return 0;
}
```

C Code Example (cont.)

- In this program:
  - Caller
    - main
  - Callee
    - pow
  - Passing parameters
    - b, e
  - Return value
    - p

Function Call

- A function call involves
  - program flow control between caller and callee
    - target/return addresses
  - value passing
    - parameters/return values
- Certain rules/conventions are used for implementing functions and function calls.

Rules (I)

- Using stack for parameter passing
- Registers can be used as well for parameter passing
  - For example, WINAVR uses
    - registers r8 ~ r25 to store passing parameters
    - r25:r24 to store the return value
  - The parameters may eventually be saved on the stack to free registers.
- Some parameters that are used in several places in the program must be saved in the stack.
  - E.g. inputs to recursive call

Rules (II)

- Parameters can be passed by *value* or *reference*
  - Passing by value
    - Pass the value of an actual parameter to the callee
      - Not efficient for structures and arrays
        - » Need to pass the value of each element in the structure or array
  - Passing by reference
    - Pass the address of the actual parameter to the callee
    - Efficient for structures and array passing
    - Using *passing by reference* when the parameter is to be modified by the function
      - Example is given in the next two slides

Rules (III)

- If a register is being used by both caller and callee and the caller needs its old value after the callee returns, then a *register conflict* occurs.
- Compilers or assembly programmers need
  - to check for register conflict
  - to save conflict registers on the stack
- Caller or callee or both can save conflict registers.
  - In WINAVR, callee saves conflict registers

Rules (IV)

- Local variables and parameters need to be stored contiguously on the stack for easy accesses.
- How are the local variables or parameters stored on the stack?
  - In the order that they appear in the high-level program from left to right, or the reverse order.
  - Either is OK. But the consistency should be maintained.
  - Example will be provided later

Stack Frame and Function Call

- Each function call creates a *stack frame* in the stack.
- The stack frame occupies some space and has an associated pointer, called *stack frame pointer*.
  - WINAVR uses **Y (r29: r28)** as the stack frame pointer
- The stack frame space is freed when the function returns.
- The stack frame pointer can point to either the base (starting address) or the top of **the stack frame**
  - In AVR, it points to the top of the stack fram



Typical Stack Frame Contents

- Return address
  - Used when the function returns
- Conflict registers
  - One conflict register is the stack frame pointer
  - The original contents of these registers need to be restored when the function returns
- Parameters
- Local variables

Stack Frame Structure: an example

```
int main(void)
{ …
  foo(arg1, arg2, …, argm);
}
 void foo(arg1, arg2, …, argm)
{ int var1, var2, …, varn;
  …
}
```



A Template for Caller

Basic operations by caller:

- Before calling the callee, store passing parameters in the designated registers
- Call callee.
  - Using instructions for function call
    - rcall, icall, call.

Relative Call to Function

- Syntax: ***rcall k***
- Operands: $-2K \leq k < 2K$
- Operation: stack ← PC+1, SP ← SP-2
  PC ← PC+k+1
- Words: 1
- Cycles: 3

- For device with 16-bit PC

A Template for Callee

Callee (function):

- Prologue
- Function body
- Epilogue

A Template for Callee (cont.)

Prologue:

- Save conflict registers, including the stack frame pointer on the stack by using *push* instruction
- Allocate space for local variables and passing parameters
  - by updating the stack pointer SP
    - SP = SP - the size of all parameters and local variables.
    - Using *OUT* instruction
- Update the stack pointer and stack frame pointer Y to point to the top of its stack frame
- Pass the actual parameters' values to the parameters' locations on the stack

Function body:

- Perform the normal task of the function on the stack frame and registers.

A Template for Callee (cont.)

Epilogue:

- Store the return value in the designated registers
- De-allocate the stack frame
  - Deallocate the space for local variables and parameters by updating the stack pointer SP.
    - SP = SP + the size of all parameters and local variables.
    - Using *OUT* instruction
  - Restore conflict registers from the stack by using *pop* instruction
    - The conflict registers must be popped in the reverse order that they were pushed on the stack.
      - The stack frame pointer register of the caller is also restored.
- Return to the caller by using *ret* instruction

Return from Subroutine Instruction

- Syntax: ***ret***
- Operands: none
- Operation: SP ← SP+2, PC ← (SP)

- Words: 1
- Cycles: 4

- For device with 16-bit PC

Example 1

- C program (power function)
  - Assume an integer takes two bytes

```
unsigned int pow(unsigned int b, unsigned int e) {        // int parameters b & e,
                                                          // returns an integer
        unsigned int i, p;                                // local variables
        p = 1;
        for (i=0; i<e; i++)                               // p = b^e
                p = p*b;
        return p;                                 // return value of the function
}

int main(void) {
        unsigned int m, n;
        m = 2;
        n = 3;
        m = pow(m, n);
        return 0;
}
```

Stack frame for pow()



Parameter Passing

- Assume an integer takes two bytes



Example 1 (cont.)

- Assembly program
  - Assume an integer takes two bytes

```
.include "m2560def.inc"
.equ m = 2
.equ n = 6

; Macro mul2: multiplication of two 2-byte unsigned numbers with a 2-byte result
; All parameters are registers, @5:@4 should be in the form: rd+1:rd, where d is
; the even number, and rd+1:rd are not r1:r0
; Operation: (@5:@4) = (@1:@0)*(@3:@2)

.macro mul2                 ; a * b
    mul   @0, @2            ; al * bl
    movw  @5:@4, r1:r0
    mul   @1, @2            ; ah * bl
    add   @5, r0
    mul   @0, @3            ; bh * al
    add   @5, r0
.endmacro
```



Example 1 (cont.)

- Assembly program

```
        ; main
        ldi r22, low(m)                 ; m = 2
        ldi r23, high(m)
        ldi r20, low(n)                 ; n = 6
        ldi r21, high(n)
        rcall pow                       ; Call function pow
        movw r23:r22, r25:r24           ; Get the return result
end:
        rjmp end                        ; end of main
```

Example 1 (cont.)

- Assembly program

```
pow:
        ; Prologue:
                                    ; r29:r28 will be used as the frame pointer
        push YL                     ; Save r29:r28 in the stack
        push YH
        push r16                    ; Save registers used in the function body
        push r17
        push r18
        push r19
        in YL, SPL                  ; Initialize the stack frame pointer value
        in YH, SPH
        sbiw Y, 8                   ; Reserve space for local variables
                                    ; and parameters.
```

Example 1 (cont.)

- Assembly program

```
out SPH, YH          ; Update the stack pointer to
out SPL, YL          ; point to the new stack top

                     ; Pass the actual parameters
std Y+1, r22         ; Pass m to b
std Y+2, r23
std Y+3, r20         ; Pass n to e
std Y+4, r21
;  End of prologue
```

Example 1 (cont.)

- Assembly program

```
; Function body
                                ; Use r23:r22 for i and r25:r24 for p,
                                ; r21:r20 temporarily for e and r17:r16 for b
clr r23;                        ; Initialize i to 0
clr r22;
clr r25;                        ; Initialize p to 1
ldi r24, 1
…                               ; Store the local values to the stack
                                ; if necessary

ldd r21, Y+4                    ; Load e to registers
ldd r20, Y+3
ldd r17, Y+2                    ; Load b to registers
ldd r16, Y+1
```

Example 1 (cont.)

- Assembly program

```
loop:   cp r22, r20                         ; Compare i with e
        cpc r23, r21
        brsh done                           ; If i >= e
        mul2 r24,r25, r16, r17, r18, r19    ; p *= b
        movw r25:r24, r19:r18
        subi r22, Low(-1)                   ; i++
        sbci r23, High(-1)
        rjmp loop
done:
        ; End of function body
```

Example 1 (cont.)

- Assembly program

```
        ; Epilogue

        adiw Y, 8           ; De-allocate the reserved space
        out SPH,  YH
        out SPL, YL
        pop r19
        pop r18                     ; Restore registers
        pop r17
        pop r16
        pop YH
        pop YL
        ret                         ; Return to main()
        ; End of epilogue
```

Recursive Function

- A recursive function
  - is both a caller and a callee of itself
  - is formed by a looped function calls
  - has a termination point or base case
- Can be hard to compute the maximum stack space needed for a recursive function call.
  - Need to know how many times the function is nested (the depth of the call).
  - And it often depends on the input values of the function
- An example is given next

Example 2

- C program (Fibonacci number function)
  - Assume an integer takes one byte

```
int n = 12;
void main(void)
 {
    fib(n);
  }

int fib(int m)
{
   if(m == 0)  return 1;
   if(m == 1)  return 1;
   return (fib(m - 1) + fib(m - 2));
}
```

Stack frame for fib()

- Assembly program
  - Assume an integer takes one byte



Parameter Passing

- Assume an integer takes one byte



Example 2 (cont.)

- Assembly program

```
.include "m2560def.inc"
.cseg
        rjmp main
n:      .db 12

main:
        ldi ZL, low(n <<1)          ; Let Z point to n
        ldi ZH, high(n <<1)
        lpm r24, Z                  ; Pass n via r24
        rcall fib                   ; Call fib(n)

halt:
        rjmp halt
```

Example 2 (cont.)

- Assembly program

```
; fib(m)
fib:                    ; Prologue
    push r16            ; Save r16 on the stack
    push YL             ; Save Y on the stack
    push YH
    in YL, SPL
    in YH, SPH
    sbiw Y, 1           ; Let Y point to the top of the stack frame
    out SPH, YH         ; Update SP so that it points to
    out SPL, YL         ; the new stack top

    std Y+1, r24        ; get the parameter
    cpi r24, 2          ; Check whether m is larger than 1
    brsh L2             ; If m!=0 or 1
    ldi r24, 1          ; m==0 or 1, return 1
    rjmp L1             ; Jump to the epilogue
```

Example 2 (cont.)

- Assembly program

```
L2:     ldd r24, Y+1        ; m>=2, load the actual parameter m
        dec r24             ; Pass m-1 to the callee
        rcall fib           ; call fib(m-1)
        mov r16, r24        ; Store the return value in r16
        ldd r24, Y+1        ; Load the actual parameter m
        subi r24, 2         ; Pass m-2 to the callee
        rcall fib           ; call fib(m-2)
        add r24, r16        ; r24=fib(m-1)+fib(m-2)
```

Example 2 (cont.)

- Assembly program

```
L1:
        ; Epilogue
        adiw Y, 1           ; Deallocate the stack frame for fib()
        out SPH, YH         ; Restore SP
        out SPL, YL
        pop YH              ; Restore Y
        pop YL
        pop r16             ; Restore r16
        ret
```

Reading Material

- AVR ATmega2560 data sheet
  - Stack, stack pointer and stack operations

Homework

1. Refer to the AVR Instruction Set manual, study the following instructions:
   - Arithmetic and logic instructions
     - sbci
     - lsl, rol
   - Data transfer instructions
     - pop, push
     - in, out
   - Program control
     - rcall
     - ret
   - Bit
     - clc
     - Sec
2. Read Introduction to AVR Microprocessor Development Board available on the Labs page.