Microprocessors & Interfacing

Interrupt (II)

Lecturer : Annie Guo

Lecture Overview

- Interrupts in AVR
  - Internal interrupt
    - Timer and timer generated interrupt

Reference: Counter* (1/2)

- A counter increases/decrease its value every clock cycle.
- Symbol



The slide and the next one were copied from
Reference: Logic Gates and Typical Functional Blocks

Reference: Counter* (2/2)

- 4-bit counter

The counter has

- a **synchronous** (同步的) **load**
- an **asynchronous** (异步的) **clear**

The counter counts through 0, 1, 2, …, 15, 0





Timer

- A timer is simply a binary counter
- Can be used to
  - Measure time duration
    - Determined by the count value and clock cycle time
  - Generate PWM signals
    - PWM: Pulse-Width Modulation
      - To be covered later
  - Schedule real-time tasks
    - Based on generated interrupts
  - Etc.

Timers in AVR

- In AVR, there are 8-bit and 16-bit timers.
  - Timer 0 and Timer 2
    - 8-bit counters
  - Timers 1, 3-5
    - 16-bit counters
- Timer0 is covered in the next slides
  - Similar designs can be found for other timers
    - See the Atmega2560 data sheet



Timer Clock Source*



8-bit Timer Block Diagram*



8-bit Timer

- The counter can be initialized with
  - 0 (controlled by reset)
  - a number (controlled by *count signal*)
- Can count up or count down
  - controlled by *direction signal*
- Those controlled signals are generated by hardware control logic
  - The control logic is further controlled by programmer by
    - Writing control bits into TCCRnA/TCCRnB

8-bit Timer (cont.)

- Outputs from the timer
  - Overflow interrupt request bit
  - Output Compare interrupt request bits,
  - OCn bits: Output Compare bit for waveform generation
- The TIMSK register is used to enable the interrupts from the timer

TIMSK0

- Timer/Counter Interrupt Mask Register for Timer0
  - Set TOIE0 (and I-bit in SREG) to enable the Overflow Interrupt
  - Set OCIE0A/B (and I bit in SREG) to enable Compare Match Interrupts



TCCR0A/B

- Timer Counter Control Register

Bit	7	6	5	4	3	2	1	0	
0x24 (0x44)	COM0A1	COM0A0	COM0B1	COM0B0	–	–	WGM01	WGM00	TCCR0A
Read/Write	R/W	R/W	R/W	R/W	R	R	R/W	R/W	
Initial Value	0	0	0	0	0	0	0	0	

Bit	7	6	5	4	3	2	1	0	
0x25 (0x45)	FOC0A	FOC0B	–	–	WGM02	CS02	CS01	CS00	TCCR0B
Read/Write	W	W	R	R	R/W	R/W	R/W	R/W	
Initial Value	0	0	0	0	0	0	0	0	

TCCR0 Bit Description

- COM0xn/WGM0n:
  - Control the mode of the timer operation
    - The behavior of the Timer/Counter and the output, is defined by the combination of the Waveform Generation mode (WGM02:00) and Compare Output mode (COM0x1:0) bits.
    - The simplest mode of operation is the Normal Mode (WGM02:00 =000). In this mode the counting direction is up. The counter rolls over when it passes its maximum 8-bit value (TOP = 0xFF) and then restarts from the bottom (0x00).
- Refer to Mega2560 Data Sheet (pages 118~194) for details.

TCCR0 Bit Description (cont.)

- Bit 2:0 in TCCR0B
  - Control the clock selection

CS02	CS01	CS00	Description
0	0	0	No clock source (Timer/Counter stopped)
0	0	1	$clk_{I/O}$/(No prescaling)
0	1	0	$clk_{I/O}$/8 (From prescaler)
0	1	1	$clk_{I/O}$/64 (From prescaler)
1	0	0	$clk_{I/O}$/256 (From prescaler)
1	0	1	$clk_{I/O}$/1024 (From prescaler)
1	1	0	External clock source on T0 pin. Clock on falling edge
1	1	1	External clock source on T0 pin. Clock on rising edge

Example 1

- Implement a scheduler that can execute a task every one second.
  - Can be realized with
    - software design,
      - Software generates the delay
        - » With nop instructions
        - » With other tasks of known execution time
    - hardware design
      - Used here and solution is given in the next slides



Example 1 Solution

- Use **8-bit** Timer0 to “count” the time
  - Let’s set Timer0 prescaler to /64 (i.e. the system frequency is divided by 64)
    - The full counting duration (time-out) for the setting should be
      - 256x(clock period) = 256x64/(16 MHz)
        = 1024 us
        - » Namely, we can set the Timer0 overflow interrupt that is to occur every 1024 us.
        - » Note, clock period = 1/16 MHz (obtained from the data sheet) the 8-bit counter can count 256 clock cycles.
    - For one second, there are
      - 1000000/1024 =~ 1000 interrupts

Example 1 Solution (cont.)

- In the assembly code,
  - Set Timer0 interrupt to occur every 1024 microseconds
    - as explained in the previous slide
  - Use a counter to count to 1000 interrupts for 1 second duration
  - To observe the 1 second time period, use LED display that toggles every 1000 interrupts (i.e. one second)
    - a dummy task that flips display pattern
  - The code is given in the next slides

Example 1

```
; This program uses Timer0 to schedule a task that occurs every second.
; The every one second is generated by Timer0 interrupts.

.include "m2560def.inc"

.equ PATTERN=0b11110000
.def temp=r16
.def leds = r17


; The macro clears a word (2 bytes) in a memory for the counter implemented in data memory
; The parameter @0 is the memory address for that word
.macro Clear
        ldi YL, low(@0)          ; load the memory address to Y
        ldi YH, high(@0)
        clr temp
        st Y+, temp              ; clear the two bytes at @0 in SRAM
        st Y, temp
 .endmacro
                                                    ; continued
```

Example 1

```
; continued
.dseg
 SecondCounter:
      .byte 2                    ; Two-byte counter for counting the number of seconds.
TempCounter:
      .byte 2                    ; Temporary counter. Used to determine
                                 ; if one second has passed (i.e. when TempCounter=1000)
 .cseg
 .org 0x0000
      jmp RESET
      jmp DEFAULT                ; No handling for IRQ0.
      jmp DEFAULT                ; No handling for IRQ1.
      ;…                         ; insert other interrupt vectors
.org OVF0addr
      jmp Timer0OVF              ; Jump to the interrupt handler for Timer0 overflow.
      ;…                         ; other default service
      jmp DEFAULT                ; default service for all other interrupts.
DEFAULT:  reti                   ; no service
                                                        ; continued
```

Example 1

```
; continued

RESET:

        ser temp                        ; set Port C as output
        out DDRC, temp

        rjmp main
                                        ; continued
```

Example 1

```
; continued

Timer0OVF:                  ; interrupt subroutine for Timer0
        ;in temp, SREG
        push temp           ; Prologue starts.
        push Yh             ; Save all conflict registers in the prologue.
        push YL
        push r25
        push r24            ; Prologue ends.
        ldi YL, low(TempCounter)    ; Load the address of  the temporary
        ldi YH, high(TempCounter)   ; counter.
        ld r24, Y+          ; Load the value of the temporary counter.
        ld r25, Y
        adiw r25:r24, 1     ; Increase the temporary counter by one.
                                                    ; continued
```

Example 1

```
; continued
        cpi r24, low(1000)          ; Check if (r25:r24)=1000
        brne NotSecond
        cpi  r25, high(1000)
        brne  NotSecond
        com leds
        out PORTC, leds
        Clear TempCounter           ; Reset the temporary counter.

        ldi YL, low(SecondCounter)  ; Load the address of  the second
        ldi YH, high(SecondCounter) ; counter.
        ld r24, Y+                  ; Load the value of the second counter.
        ld r25, Y
        adiw r25:r24, 1             ; Increase the second counter by one.

                                                    ; continued
```

Example 1

```
; continued
        st Y, r25               ; Store the value of the second counter.
        st -Y, r24
        rjmp endif
NotSecond:
        st Y, r25               ; Store the value of the temporary counter.
        st -Y, r24
endif:
        pop r24                 ; Epilogue starts;
        pop r25                 ; Restore all conflict registers from the stack.
        pop YL
        pop YH
        pop temp
        ;out SREG, temp
        reti                    ; Return from the interrupt.

                                                        ; continued
```

Example 1

```
; continued

main:
        ldi leds, 0xff              ; Init pattern displayed
        out PORTC, leds
        ldi leds, PATTERN
        Clear TempCounter           ; Initialize the temporary counter to 0
        Clear SecondCounter         ; Initialize the second counter to 0
        ldi temp, 0b00000000
        out TCCR0A, temp
        ldi temp, 0b00000011
        out TCCR0B, temp            ; Prescaler value=64, counting 1024 us
        ldi temp,  1<<TOIE0
        sts TIMSK0, temp            ; T/C0 interrupt enable
        sei                         ; Enable global interrupt
  loop:
        rjmp loop                   ; loop forever
```

Reading Material

- Chapter 10: Interrupts and Real-Time Events. Microcontrollers and Microcomputers by Fredrick M. Cady.
- Mega2560 Data Sheet.
  - External Interrupts.
  - Timer0

Homework

1. An underground oil tank monitor system has the following functions:
   1. read(): to read the tank oil level
   2. display(): to display the oil level
   3. main(): process a few of basic tasks: if the oil level is below the low limit, do something; if oil level is over the high limit, do something else; and other routine work.

   It is required that the display should be updated every 1 minute, reading should be done at least every 10 seconds. Assume read() and display() take 1 ms and 5 ms, respectively. Design a scheduling controller for those functions so that the above requirements can be met, and the design leads to an easy assembly code implementation.